

ESERCITAZIONE DI MODELLAZIONE SOLIDA

L' esercitazione, suddivisa in due parti, consiste nel realizzare il dispositivo di aggancio a "c" mostrato in Fig.1. Nella prima parte viene creata una mesh non strutturata ("free"), mentre nella seconda ne viene creata una strutturata ("mapped"). L'esercitazione ha lo scopo principale di fornire un esempio di applicazione di alcuni comandi rilevanti. Si sottolinea, anche in questo caso, come la strategia di costruzione del modello proposta non debba essere considerata l'unica possibile.

Parte 1 – Free meshing

La costruzione della geometria segue la tecnica detta 'top down': si ottiene cioè il modello solido attraverso l'uso di primitive geometriche e operazioni booleane di 'sculpturing'. Il solido effettivamente modellato corrisponde ad un quarto del corpo reale, grazie alla doppia simmetria di quest'ultimo.

I volumi delle primitive costituiscono entità geometriche disgiunte, anche se sovrapposte, sulle quali è necessario operare sia per ottenere il solido desiderato (realizzazione del foro) che per ottenere un solido connesso sul quale realizzare la mesh.

L' operazione booleana di addizione (VADD) permette di unire i due parallelepipedi in un unico volume privo di superfici d'interfaccia, mentre l'operazione di overlap (VOVLAP) permette di generare entità geometriche distinte ottenute dalle intersezioni delle entità geometriche originarie più le entità complementari alle intersezioni stesse, con superfici d'interfaccia in comune; questo permette di definire il volume da rimuovere per realizzare il foro e nello stesso tempo di generare volumi distinti e collegati utilizzabili per la realizzazione della mesh.

Tramite il comando LSEL è possibile selezionare le linee su cui imporre le divisioni volute per la meshatura (LESIZE). Il comando SMRTSIZE permette di attivare una procedura automatica di suddivisione che, partendo dalle dimensioni fissate tramite l' ESIZE, e rispettando le divisioni imposte esplicitamente con LESIZE, infittisce automaticamente la mesh su bordi curvi e in prossimità di spigoli o superfici, controllando l' espansione degli elementi passando dai bordi all'interno delle aree, nonché altri fattori legati alla meshatura.

L' elemento utilizzato per la mesh free deve essere obbligatoriamente un tetraedro (solid72 o solid92) e la mesh stessa deve essere richiesta tramite il comando MSHKEY,0.

Il modello ad elementi finiti completo è ottenuto per riflessione (NSYM ed ESYM) sui piani di simmetria del solido. La continuità del modello FEM è garantita da una operazione di unione dei nodi alle interfacce (piani di simmetria) con il comando NUMMRG. Preliminarmente, vengono ottenute alcune informazioni sul modello con *GET.

Si ottiene in tal modo un modello solido che rappresenta 1/4 del corpo ed un modello FEM completo.

Parte 2 – Mapped meshing

La modellazione solida segue modalità analoghe a quelle impiegate nella Parte 1. Tuttavia, il requisito di creare una mesh di tipo mapped impone di ottenere una suddivisione del modello solido in sottovolumi di forma opportuna, ciascuno dei quali in possesso dei seguenti requisiti:

- (1) il volume deve essere delimitato da 6 aree (tipo parallelepipedo), da 5 aree (tipo prisma o cono) oppure da 4 aree (tetraedro)
- (2) deve avere un egual numero di divisioni su spigoli opposti (a meno di non rispettare schemi particolari riportati sul manuale)
- (3) il numero di divisioni sugli spigoli delle facce di forma triangolare deve essere pari se il volume è un prisma o un tetraedro.

A tael scopo è necessario operare sulle primitive generate.

Il cilindro ed il parallelepipedo a questo adiacente vengono divisi in due parti con il working pPlane (VSBW: sottrazione del WP dai volumi), e successivamente viene realizzata una operazione di overlap (VOVLAP) su tutti i volumi presenti. E necessario, inoltre, concatenare alcune aree (ACCAT) appartenenti alle parti di parallelepipedo intersecate dal cilindro. La concatenazione delle linee relative alle aree concatenate viene effettuata automaticamente.

Queste operazioni permettono di realizzare il foro e, nello stesso tempo, di preparare dei volumi sui quali possa realizzarsi la mesh, ottenuta utilizzando elementi brick (solid45 o solid95), ed imponendo al sistema di generare una mesh mapped con il comando MSHKEY,1.

Il modello FEM finale è ottenuto per riflessione dei volumi sui rispettivi piani di simmetria (VSYM), operazione che implica anche la riflessione della mesh associata al volume originale. Il modello è infine reso continuo attraverso una operazione di unione dei nodi d'interfaccia (NUMMRG).




Figura 1 – Corpo solido oggetto dello studio




Fig. 2 – Free Meshing




Fig. 3 – Mapped Meshing

*Progettazione assistita delle strutture meccaniche
Esercitazione di modellazione solida Parte 1*

```
FINISH
/CLEAR
/TITLE,MODELLAZIONE-1
C***
C*** DEFINIZIONE DI ALCUNI PARAMETRI DI MODELLAZIONE
T=2          !!SPESSORE PARALLELEPIPEDI
L1=10
L2=10
L3=20
HT=10
RI=2.5       !!RAGGIO INTERNO CILINDRO
RE=5         !!RAGGIO ESTERNO CILINDRO
SIZL=8       !!SIZE LEVEL PER SMARTSIZE (1-10=FINE-COARSE) USARE ;?:.8
C***
C*** DEFINIZIONE ELEMENTI
C***
/PREP7
C***
C*** ELEMENTI
C***
ET,1,72      !!ELEMENTO TETRAEDRO A 4 NODI CON 6 DOF PER NODO (SOLID72)
C***
C*** PROPRIETA' MATERIALE
C***
MP,EX,1,210000  !! ACCIAIO:MODULO DI YOUNG 210000 MPA
MP,NUXY,1,0.3  !! COEFFICIENTE DI POISSON 0.3 (NON NECESSARIO)
C***
C*** GENERAZIONE DEL MODELLO SOLIDO
C***
BLC5,-L1/2,T/2,L1,T,L2    !! PARALLELEPIPEDO CON CENTRO, LARGHEZZA, ALTEZZA
BLC4,-L1,0,T,L3,L2      !! PARALLELEPIPEDO CON VERTICE, SECONDO VERTICE E PROF.
WPOFFS,-L1,HT,L2        !! OFFSET DEL WORKING PLANE
WPROTA,,90              !! ROTAZIONE DEL WP SU Y DI 90 GRADI
CYL4,0,0,RE,90,RI,-90,L1  !! COSTRUZIONE DEL CILINDRO CAVO
VADD, 1,2               !! UNISCE I DUE PARALLELEPIPEDI
VOVLAP,3,4              !! OVERLAP PER GENERARE I VOLUMI DI INTERSEZIONE
CSWPLA, 11, 1           !! RIF. CILINDRICO ALLINEATO AL WP
C***
C*** SELEZIONE DEL VOLUME DA RIMUOVERE
C***
ASEL,S,LOC,X,0,RI-0.1    ! SELZIONE AREE
VSLA                    !!SELEZIONA IL VOLUME COLLEGATO ALLE AREE
VDEL,ALL,,1            !! RIMOZIONE DEL VOLUME
ALLS
C***
C*** DIVISIONI DIVERSE PER LE DIVERSE LINEE
C***
ESIZE,NINT(T/2)         !!DIMENSIONE DI PARTENZA PER SMRTSIZE
KSEL,S,LOC,X,RE-0.1,RE+0.1  ! SELEZIONE KEYPOINTS
KSEL,R ,LOC,Z,T-0.1 ,L1 +0.1
LSLK,S,1                !!SELEZIONA LE LINEE COMPLETAMENTE RACCHIUSE DAI KPS ATTIVI
LSEL,R ,LOC,Y , -80,80
LESIZE,ALL,,9,,1 ,1     ! !IMPONE UNA DIVISIONE DELL 'ARCO (9 GRADI)
ALLS
LSEL,S,LOC,X,RE-0.1,RE+0.1
LSEL,R ,LOC,Z,T+0.1 ,L1-0.1
LESIZE,ALL,,10,0.25,1  !!IMPONE UNA DIVISIONE 'A SCALARE' DALLA BASE VERSO IL CENTRO
LSEL,S,LOC,X,RI-0.1,RI+0.1
LSEL,R ,LOC,Z,T+0.1 ,LI-0.1
LESIZE,ALL,,10,4,1     ! !IMPONE UNA DIVISIONE.'A SCALARE' DALLA BASE VERSO IL CENTRO
```

```

ALLS
C***
C*** OVERLAP GARANTISCE LA CONTINUITÀ DEL MODELLO ALLE INTERFACCE POICHÈ
C*** I VOLUMI GENERATI HANNO SUPERFICI D'INTERFACCIA IN COMUNE
C***
C*** GENERAZIONE DELLA MESH
C***
SMRTSIZE,SIZL          ! !SI ATTIVA LO SMARTSIZE
MSHAPE,L
MSHKEY ,0              !IMPONE AL SISTEMA DI GENERARE UNA MESH NON STRUTTURATA
VMESH,ALL
C***
C*** GENERAZIONE DEL MODELLO FEM COMPLETO SENZA COPIA DEI VOLUMI
C***
LOCAL,12,0,0,0,10      !!DEFINISCE ED ATTIVA UN RIFERIMENTO CARTESIANO AUSILIARIO
*GET,NMX,NODE,,NUM,MAX  !! METTE NELLA VARIABILE NMX IL MASSIMO NUMERO DI NODO
NSYM,X,NMX,ALL         ! RIFLESSIONE DEI NODI
ESYM,,NMX,ALL          ! RIFLESSIONE ELEMENTI
*GET,NMX,NODE,,NUM,MAX ! NUOVA RIFLESSIONE DELL'INTERO MODELLO
NSYM,Z,NMX,ALL
ESYM,,NMX,ALL
C***
C***MODELLO FEM COMPLETO; MODELLO SOLIDO: UN QUARTO DEL VOLUME TOTALE
C*** MERGE DEI SET COPIATI PER RENDERE CONTINUO IL MODELLO ALLE INTERFACCE
C***
NUMMRG,NODE
NUMCMP,NODE  ! compatta i numeri di nodi
/VIEW,L,L,L,L
EPLO

```

*Progettazione assistita delle strutture meccaniche
Esercitazione di modellazione solida Parte 2*

```
FINISH
/CLEAR
/TITLE,MODELLAZIONE-2
C***
C*** DEFINIZIONE DI ALCUNI PARAMETRI DI MODELLAZIONE
C***
T=2
L1=10
L2=10
L3=20
HT=10
RI=2.5
RE=5
SIZEL=2
C***
C*** DEFINIZIONE ELEMENTI
C***
/PREP7
C***
C*** ELEMENTI
C***
ET,1,45                !! ELEMENTO 1: SOLID45, BRICK AD 8 NODI CON 3 DOF PER NODO
C***
C*** PROPRIETA' MATERIALE
C***
MP,EX,1,2L0000        !! ACCIAIO:MODULO DI YOUNG 210000 MPA
MP,NUXY,1,0.3        !! COEFICIENTE DI POISSON 0.3
C***
C*** GENERAZIONE DEL MODELLO SOLIDO
C***
BLC5,-L1/2,T/2,L1,T,L2    !! PARALLELEPIPEDO CON CENTRO, WIDTH, HEIGHT
BLC4,-L1,0,T,L3,L2      !! PARALLELEPIPEDO CON VERTICE, SECONDO VERTICE E PROF.
WPOFFS,-L1,L2,HT        !! OFFSET DEL WORKING PIANE
WPROTA,,90              !! ROTAZIONE DEL W P SU Y DI 90 GRADI
CYL4,0,0,RE,90,RI,-90,L1  !! COSTRUZIONE DEL CILINDRO CAVO
WPROTA,,90              !! ROTAZIONE W P SU X DI 90 GRADI
VSBW,2
VSBW,3                  !! TAGLIO DEI VOLUMI CON IL WORKING PIANE;
VOVLAP,ALL              !! OVERLAP PER GENERARE I VOLUMI DI INTERSEZIONE
C***
C*** SELEZIONE DEL VOLUME DA RIMUOVERE
C***
KBOS=KP(-L1,L2,HT)      !! INDIVIDUA IL KP CENTRO DEL CERCHIO DI BASE DEL VOLUME
KSEL,S,,,KBOS
LSLK
ASLL
VSLA
VDEL,ALL,,,1 !! RIMOZIONE DEL VOLUME
ALLS
C***
C*** GENERAZIONE DELLA MESH STRUTTURATA
C*** PREPARAZIONE DEL SOLIDO PER LA MESH STRUTTURATA
C*** CONCATENAZIONE DELLE AREE PER LE PARTI FORATE
C***
ASEL,S,LOC,Y ,L3-0.1 ,L3+0.1
ASEL,A,LOC,Z,-0.1,0.1
ASEL,R ,LOC,Y ,L3/2,L3
ACCAT,ALL                !!CONCATENAZIONE DELLE AREE PER AVERE UNO SCHEMA PER IL
VOLUME
```

!!IN CUI SIA POSSIBILE COSTRUIRE UNA MESH MAPPED

```
ALLS
ASEL,S,LOC,Z,-0.1,+0.1
ASEL,R,LOC,Y,T,L3/2
ASEL,A,LOC,Y,T-0.1,T+0.1
ASEL,R,LOC,X,-L1+0.1,-L1+T-0.1
ACCAT,ALL
ALLS
C***
C*** GENERAZIONE DELLA MESH STRUTTURATA (MAPPED)
C***
MSHKEY,1                !!IMPONE AL SISTEMA DI GENERARE MAPPED MESH
MSHAPE,0,3D
ESIZE,SIZEL             ! !ELEMENTO =5MM DI LATO SULLE LINEE DEL SOLIDO
VMESH,ALL              ! !MESH DEI VOLUMI
C***
C*** GENERAZIONE PER SIMMETRIA DEL RESTO DEL MODELLO FEM
C***
ASEL,S,ACCAT
ADEL,ALL
LSEL,S,LCCAT
LDEL,ALL
ALLS
*GET,NL, VOLU,,NUM,MAX
NST=NL+1
NUMSTR, VOLU,NST
VSYM,X,ALL,,,0,0      ! !COPIA VOLUMI ED ELEMENTI PER SIMMETRIA SU YZ
LOCAL, 11,0,0,0,10   ! !DEFINISCE ED ATTIVA UN RIFERIMENTO CARTESIANO AUSILIARIO
VSYM,Z,ALL,,,0,0     ! !GENERA I VOLUMI RIMANENTI PER SIMMETRIA SU XY
NUMMRG,NODE          ! !MERGE DEI NODI PER RENDERE CONTINUO IL MODELLO
NUMCMP,NODE          ! !COMPRIE I NUMERI DEI NODI
/VIEW,L,1,1,1        !!SI VISUALIZZA IL MODELLO FEM COMPLETO IN PROPSETTIVA
EPLO
```